



It's cool to see how compact is the code!

char=CharacterRange[" ","~"]~Join~Map[Style[#,Bold]&,CharacterRange[" ","~"]];

w=N@Mean@Flatten@ImageData@Binarize@Rasterize@#->#&/@char;

w[[All,1]]=#~Rescale~{Min[#],Max[#]}&@w[[All,1]];

findChar=Nearest[w];



asciiIt[image_]:=Module[{img},

img=ImageAdjust@ColorConvert[image,"Grayscale"];

img=ImageData@ImageResize[img,80];

img=Map[First@findChar[#,1]&,img,{-1}];

Grid[img,Spacings->{0.1,-0.2},ItemSize->{0.4,0.4}]

]



Dynamic[asciiIt@CurrentImage[]]

So we get:



Inspired on this site I tried to create a Wolfram Language version of an asciicam.It's cool to see how compact is the code!So we get: